******************************
* Stata preamble
******************************
clear all
macro drop _all
capture log close
set linesize 255
set mat 1500
set more off,perm

				
cd /*set to Directory with BW_RCFS_Replication_Dataset.dta*/
capture log close
log using log_01_RCFS_Main_Tables_and_Figures.txt, replace text
use BW_RCFS_Replication_Dataset.dta , clear


************************************************************
* Table 1: Summary Statistics
************************************************************	
eststo clear
qui eststo SUM_STATS: estpost  tabstat covid_cases ///
	ev_outn95masks ev_outsurgicalmasks ev_outeyeprotection cum_outn95masks_10_18_2020 cum_outsurgicalmasks_10_18_2020 cum_outeyeprotection_10_18_2020	///
	Mcare_share Mcaid_share Mcaid_Boost Cash_1k logCash DCoH Cash_beds1k Cash_NetRev ///
	county_covid_nonSNF_1k ResChangeJanJune AvgResidents Beds admissions_hundreds ///
	TotalMargin Leverage ForProfit FacilityAge LPNhrsPerResDay RNhrsPerResDay ///
	HealthInspRating ZipMinorityShare ZipHousholdIncome_1k SqFtperResident AveUniqueVisitors	 ,stats(mean sd min p25 median p75 max n ) c(s)
esttab SUM_STATS, cells("mean(fmt(%8.2fc)) sd  p25 p50 p75  count") varwidth(32) label
			
************************************************************
* Table 2: Correlation of Nursing Home Characteristics
************************************************************	
eststo clear
qui eststo CORR: qui estpost corr  Cash_1k DCoH  Cash_beds1k Cash_NetRev Mcare_share Mcaid_share ///
			AvgResidents Beds admissions_hundreds TotalMargin Leverage HealthInspRating   , matrix   
esttab CORR, unstack varwidth(32) label not b(2) nostar title(Correlation in Nursing Home Characteristics)

************************************************************
* Table 3: Baseline Drivers
************************************************************
global REGTABLE b(%10.3f) se(%8.3f) scalars(N alpha ll  aic bic  "depvar Dependent Var") star(* 0.10 ** 0.05 *** 0.01)   ///
	addnote("`c(username)' ran on $S_DATE $S_TIME")  varwidth(32) noconst compress label
eststo clear
 qui {
	eststo, prefix(nb_): nbreg  covid_cases z_county_covid_nonSNF 		 	       												,  cluster(CntyFE)
	eststo, prefix(nb_): nbreg  covid_cases    			         														i.CntyFE,  cluster(CntyFE)
	eststo, prefix(nb_): nbreg  covid_cases z_Beds   				 z_AvgResidents    									i.CntyFE,  cluster(CntyFE)
	eststo, prefix(nb_): nbreg  covid_cases z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3	i.CntyFE,  cluster(CntyFE)
	eststo, prefix(nb_): nbreg  covid_cases z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge z_ZipBlackShare z_LPNhrsPerResDay z_RNhrsPerResDay i.HealthInspRating i.CntyFE,  cluster(CntyFE)
 }
esttab nb*  , eform $REGTABLE ///
	title(Baseline Drivers) mtitles  order(z_county_covid_nonSNF z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay ) ///
	indicate("Rating FE =*HealthInsp*" "Cnty FE = *Cnty*"  )

************************************************************
* Figure 1: Medicare Share and Falling Occupancy
************************************************************
binscatter  pc_ResChangeJanJune   Mcare_share ,  name(ChangeRes, replace) legen(pos(6) row(1) size(tiny)) absorb(CntyFE ) ///
	control( z_logCash  z_Mcaid_share	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating ) ytitle("% Change in Residents (January-June 2020)")  nquantiles(14)  xtitle(Pre-Pandemic Medicare Share) 

************************************************************
* Figure 2: County COVID-19 Cases and COVID-19 in Nursing Homes
************************************************************
binscatter ln_covid_cases ln_county_covid_nonSNF, name(logvsNonSNF_County, replace) ytitle("log(Nursing Home Covid Cases)") xtitle("log(County Cases excluding Nursing Homes)")
	
************************************************************
* Table 4: Financial Factors and COVID-19
************************************************************
eststo clear
qui {
	eststo,prefix(nb_): nbreg  covid_cases   z_logCash	 		z_Mcare_share z_Mcaid_share	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
	eststo,prefix(nb_): nbreg  covid_cases   z_DCoH	z_Mcare_share z_Mcaid_share	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
	eststo,prefix(nb_): nbreg  covid_cases   z_Cash_Beds	 	z_Mcare_share z_Mcaid_share	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
	eststo,prefix(nb_): nbreg  covid_cases   z_Cash_NetRev	 	z_Mcare_share z_Mcaid_share	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
	eststo,prefix(nb_): nbreg  covid_cases   z_logCash	 		z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
}
esttab nb_*   , eform $REGTABLE title(Financial Factors and COVID-19) mtitles  order(z_logCash z_DCoH z_Cash_Beds z_Cash_NetRev z_Mcare_share z_Mcaid_share *Mcaid_Boost* ) ///
	indicate("Facility Controls = z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay" "Rating FE =*HealthInsp*" "Cnty FE = *Cnty*" ) 
	
************************************************************
* Figure 3: Medicaid Boost with 90% CI
************************************************************
qui nbreg  covid_cases   logCash  Mcare_share   Mcaid_share c.Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_admissions_hundreds z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE  ,  cluster(CntyFE)
	su Mcaid_share,d
	margins ,  dydx() at(Mcaid_Boost = (0 1) Mcaid_share=(0.07 0.62 0.82))
	marginsplot ,level(90) name(Mcaid_Boost,replace) legend(pos(6) row(1) order(- "Medicaid Share:" 1 "p10 (7% Medicaid Share)" 2 "p50 (62%)" 3 "p90 (82%)")) xtitle("Medicaid Boost") ytitle(Predicted number of cases) title("")

************************************************************
* Table 5: Robustness
************************************************************
eststo clear
qui {
	eststo,prefix(nbrob_): nbreg  covid_cases z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost w_ave_HCC w_pac_average_age  w_pac_pc_Male w_pac_pct_pac_lt_7days w_pac_pct_pac_lt_30days w_pac_pct_pac_gt_60days w_pac_pc_McareAdvantage w_pac_pc_DualBenef w_pac_pc_Afib w_pac_pc_Asthma w_pac_pc_CHF w_pac_pc_COPD w_pac_pc_PrimCOPD w_pac_pc_PrimRespFailure w_pac_pc_PrimCHF w_pac_pc_PrimInfection w_pac_pc_PrimInjury  z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit   i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
	eststo,prefix(nbrob_): nbreg  covid_cases z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost ln_ziphouseincome z_SqFt_per_Resident  ln_visitors_DecMar z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit   i.HealthInspRating i.CntyFE  ,  cluster(CntyFE)
	eststo,prefix(nb_CBSA_St): nbreg  covid_cases   z_logCash z_county_covid_nonSNF	z_Mcare_share z_Mcaid_share  c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CBSA_St ,  cluster(CBSA_St)
	eststo poi_0: poisson  covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,  cluster(CntyFE)
}
esttab ,eform $REGTABLE title([COVID Date=$COVID_DATE] PrCovid - Cash Flow Shocks: NegBinomial ) mtitles  order(z_logCash  z_Mcare_share z_Mcaid_share z_county_covid_nonSNF ) ///
	indicate("Facility Controls = z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay " ///
	"CaseMixDemos = w_ave_HCC w_pac_average_age  w_pac_pc_Male w_pac_pct_pac_lt_7days w_pac_pct_pac_lt_30days w_pac_pct_pac_gt_60days w_pac_pc_McareAdvantage w_pac_pc_DualBenef w_pac_pc_Afib w_pac_pc_Asthma w_pac_pc_CHF w_pac_pc_COPD w_pac_pc_PrimCOPD w_pac_pc_PrimRespFailure w_pac_pc_PrimCHF w_pac_pc_PrimInfection w_pac_pc_PrimInjury" ///
	"Rating FE =*HealthInsp*" "Cnty FE = *Cnty*"   "CBSA_St FE = *BSA*" ) 

************************************************************
* Figure IA.1: Model Fit: Negative Binomial versus Poisson
************************************************************
qui {
	poisson  covid_cases     ,  nolog
		mgen, stub(poisPDF) pr(0/30) meanpred
	poisson  covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE,nolog
		mgen, stub(PRM) pr(0/30) meanpred

	nbreg  covid_cases      ,  nolog
		mgen, stub(nbinPDF) pr(0/30) meanpred
	nbreg covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,nolog
		mgen, stub(NBRM) pr(0/30) meanpred
}	
lab var poisPDFobeq "Observed"
lab var PRMpreq "Poisson Regression fit"
lab var PRMval "Cases"
lab var NBRMpreq "Negative Binomial Regression fit"
list PRMval poisPDFobeq poisPDFpreq PRMpreq nbinPDFpreq NBRMpreq  in 1/31
 
graph twoway connected poisPDFobeq PRMpreq NBRMpreq PRMval if PRMval<=25, ytitle("Probability")  xlabel(0/25)  name(ModelFit,replace) ///
		legend(pos(6) row(1))

************************************************************
* Table 6: Personal Protective Equipment Shortages
************************************************************
eststo clear
qui foreach v in n95masks surgicalmasks eyeprotection   {
	eststo, prefix(nb`v'): nbreg cum_out`v'_10_18_2020  z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE  ,  cluster(CntyFE)
	eststo, prefix(d`v'): reg	 ev_out`v'   			z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE ,   cluster(CntyFE)
}
esttab nb*  ,eform $REGTABLE  title(Cumulative Shortages [Negative Binomial] ) mtitles  order(z_logCash  z_Mcare_share z_Mcaid_share  ) ///
	indicate( "Facility Controls =z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" "Rating FE =*HealthInsp*" "Cnty FE = *Cnty*" ) 
esttab d* , $REGTABLE  title(Any Shortages [LPM]) mtitles  order(z_logCash  z_Mcare_share z_Mcaid_share   ) ///
	indicate( "Facility Controls =z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" "Rating FE =*HealthInsp*" "Cnty FE = *Cnty*"  ) 

************************************************************
* Table IA.1: Financial Factors and COVID-19: Alternative Dates
************************************************************
eststo clear
qui foreach y in Sept Nov   {
	eststo `y'_1: nbreg covid_cases_`y'   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE     $FILT    ,  cluster( 	CntyFE  )
	eststo `y'_2: nbreg covid_cases_`y'   z_DCoH	z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE     $FILT    ,  cluster( 	CntyFE  )
	eststo `y'_3: nbreg covid_cases_`y'   z_Cash_Beds z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE     $FILT    ,  cluster( 	CntyFE  )
	eststo `y'_4: nbreg covid_cases_`y'   z_Cash_NetRev z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE     $FILT    ,  cluster( 	CntyFE  )
 }
 esttab, eform  $REGTABLE title(Financial Factors and COVID-19: Alternative Dates) mtitles  order(z_logCash z_DCoH z_Cash_Beds z_Cash_NetRev z_Mcare_share z_Mcaid_share  ) ///
			indicate( "Facility Controls = z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" ///
			"Rating FE =*HealthInsp*" "Cnty FE = *Cnty*" ) 
		

************************************************************
* Table IA.2: Personal Protective Equipment Outages: Alternative Measures of Cash
************************************************************
foreach c in z_logCash z_DCoH z_Cash_Beds z_Cash_NetRev {
eststo clear
	qui foreach v in n95masks surgicalmasks eyeprotection  {
		eststo, prefix(nb`v'): nbreg cum_out`v'_10_18_2020  `c' z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE  ,  cluster(CntyFE)
		eststo, prefix(d`v'): reg ev_out`v'   				`c'	z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE,   cluster(CntyFE)
	}
	esttab nb*  ,eform $REGTABLE  title(Cumulative Shortages [Negative Binomial] Cash Measure: `c' ) mtitles  order(`c'   ) ///
		indicate( "Facility Controls =*Mc* z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" "Rating FE =*HealthInsp*" "Cnty FE = *Cnty*" ) 
	esttab d* , $REGTABLE  title(Any Shortages [LPM] Cash Measure: `c') mtitles  order(`c'  ) ///
		indicate( "Facility Controls =*Mc* z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" "Rating FE =*HealthInsp*" "Cnty FE = *Cnty*"  ) 
}	

************************************************************
* Table IA.3: Alternative Ratings, Foot-traffic, and ICU Capacity
************************************************************
qui {
eststo clear
	eststo : nbreg covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.overall_rating_2020_01 i.CntyFE,  cluster(CntyFE)
	eststo : nbreg covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost log_AveUniqueVisitors6mo	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE,  cluster(CntyFE)
	eststo : nbreg covid_cases   z_logCash z_Mcare_share z_Mcaid_share c.z_Mcaid_share#i.Mcaid_Boost	z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit i.HealthInspRating i.CntyFE if topHosp_ICUBeds <=85   ,  cluster(CntyFE)
}
esttab ,eform $REGTABLE title(Robustness[SNF Ratings] PrCovid - Cash Flow Shocks: NegBinomial ) mtitles   ///
	indicate("HealthInsp Rating FE =*HealthInsp*" "Overall Rating FE =*rall_rat*" "Facility Controls = z_Beds z_Beds_2 z_Beds_3 z_AvgResidents z_AvgResidents_2 z_AvgResidents_3 z_FacilityAge  z_ZipBlackShare  z_LPNhrsPerResDay z_RNhrsPerResDay z_TotalMargin z_Leverage ForProfit" "Cnty FE = *Cnty*" ) 

log close
